
Introduction

Class modules are templates for objects defined by the Visual Basic programmer. A class module defines the properties and methods for an object, then allows you to create objects from that template.

A Word About Objects

You use Class modules to do Object Oriented Programming (sometimes abbreviated OOP). The word object is a big buzzword in the computer press these days; there are object oriented programming languages, object oriented databases, object oriented operating systems, object this, and object that. What does object mean? It may help if you replace the word object with another word: the word thing. In VBA, an object is a thing that exists as a memory construct that you can interact with through its properties, methods, and events. The object may have a visual interface, such as forms and controls. It may also be purely code, with no visual interface, such as an interface to a database. Objects created with class modules do not have a visual interface, whereas objects created with form modules do.

Terminology

Various terms are used in object oriented programming:

Table 1: Object Oriented Programming Terminology
Term
Meaning

Class
A template for an object. The class defines the interface for the object through its properties, methods, and events. A class module defines a class in VBA. A class does not consume memory; it is just a directive to the VBA compiler telling it that you can create objects with the specified attributes.

Object
A memory construct constructed from a class. In VBA, you interact with an object through the properties, methods, and events defined in its class module or form.

Instantiation
The process of creating an object from its class. The constructor is called when the object is instantiated.

Destruction
The process of destroying an object. The destructor is called when the object is destroyed.

Encapsulation
All of the code implements the object must be contained within the module that defines the object.

Inheritance
The process of having one class derived from and having similar attributes to another class. VBA has n6 built-in support for inheritance.

Polymorphism
Several different kinds of objects which have a common interface.

Constructor
A procedure that is called automatically when the object is instantiated where you can initialize the default values for the object. In VBA, this is the Initialize event.

Destructor
A procedure that is called automatically when the object is destroyed where you can clean up. In VBA, this is the Terminate event.

Properties
The description of the object.

Methods
A procedure that tells the object what to do.

Events
Procedures that are called when specific things happen to the object.

Advantages and Disadvantages of Object Oriented Programming

Object oriented programming was invented because of failures of large software projects. Most programmers can make a 100-line program relatively bug-free. However, few programmers can make a 10,000-line program bug-free. Several factors define a good program. A good program is (Meyer, 1988):

· Correct

The ability of the software to exactly perform their tasks, as defined by the requirements and specifications.

· Robust

The ability of the software to function even in abnormal conditions.

· Extendable

The ease with which the software may be adapted to changes of specifications.

· Reusable

The ability of the software to be reused, in whole or in part, in new applications.

· Compatible

The ease with which the software may be combined with other software.

· Efficient

How well the software makes good use of hardware resources.

· Portable

The ease with which the software may be transferred to other hardware and software platforms.

· Verifiable

The ease of preparing acceptance procedures, particularly test data, and procedures for detecting failures and tracing them to errors during the validation and operation phases.

· Secure

The ability of the software to protect its various components (program, data, and documents) against unauthorized access and modification.

· Easy to Use

The ease of learning how to use the software, operate it, prepare input data, interpret results, and recover from usage errors.

Object oriented programming helps with correctness, robustness, extendibility, reusability, and compatibility. Object oriented programming does not help with efficiency, portability, verifiability, security, and ease of use. Object oriented programming may even make implementing the code more difficult for these last items.

When To Use Class Modules

There are two good reasons for using Class Modules in VBA:

1. To allow multiple instances of a class.

Data structures are representations of the organization of data.

2. As the interface to ActiveX components.

ActiveX components are ways of breaking up the application into independently compilable and distributable files.

You can also use a class module in most places that you use a standard module, then instantiate a single instance of that object to interact with. However, there is no compelling reason to do so unless you intend to use the code in an ActiveX component.

Data Structures

Data structures are ways of representing the relationship between pieces of data. There are an infinite number of different kinds of data structures, but a few are extremely common. This section looks at three of them:

· Linked List

· Doubly Linked List

· Binary Tree

Linked List

A linked list starts with a pointer to the first object in the linked list. Each object in the linked list points to the next object in the list. The last object in the list points at nothing.

Figure 1: Linked List

The advantage of a linked list is insertion. To insert into the list, you only need to create a new object and fix up some pointers.

Figure 2 : Inserting Into A Linked List
Doubly Linked List

The disadvantage of a linked list is that you can only traverse to the next object in the list. To allow you to traverse back to the previous object in the linked list you can use a doubly linked list.

Figure 3: Doubly Linked List

Insertion works the same as a normal linked list, except there are more pointers to fix up.

Binary Tree

While insertion is easy with linked lists, searching must always be sequential. It is easy to insert and binary search binary trees. Each object in the tree can point at two others: an object that comes before it and another that comes after.


[image: image4.wmf]
Figure 4: Binary Tree

A balanced binary tree allows you to find an item traversing a maximum of Log2N number of objects, where N is the total number of objects in the binary tree. The problem with binary trees is keeping them balanced. Unbalanced trees devolve into structures almost like linked lists.

Figure 5: Unbalanced Binary Tree

There are algorithms to rebalance a binary tree. You may also use other more complicated data structures such as B-Trees (not discussed in this article) that have the advantages of binary trees but are self-balancing.

Implementing a Stack Using Class Modules

A cafeteria plate dispenser is the model for the stack. In a cafeteria plate dispenser, you can:

· Take (pop) the top plate off the stack

· Push a new plate onto the stack

· See if the stack is empty

· Look at the top plate


[image: image6.wmf]
Figure 6: A Cafeteria Plate Dispenser

A stack is an abstract data structure because the idea of the stack is independent of its implementation.

There are many ways of implementing a stack. You could use any of the following methods to actually store the plates on the stack, as well as some others:

· A dynamic array

· A collection

· A table in a database

· A text file written to disk

· A linked list

This article implements the internals of our stack using a linked list and demonstrates the use of class modules to construct data structures. Later, it shows how to encapsulate the stack into another class module. Finally, it shows how to produce an ActiveX component from the stack. The ActiveX component pulls the entire functionality of the stack into a separate file that is independently maintainable and distributable.

You only have to worry about inserting and removing the first item in the linked list when you use a linked list to implement a stack.

Creating the Interface

To start with, create a user interface. Start Visual Basic and create a Standard EXE. It gives you a form to start with.

Select Project Properties, and set the Project Name to “StackUI”. Then press OK.

View the Properties Window and set the properties for the form to:

Property
Value

Name
frmStack

Border Style
3-Fixed Dialog

Caption
Stack Example

Height
1500

StartUpPosition
CenterScreen

Width
3120

Then create a textbox and set the properties to:

Property
Value

Name
txtPlateType

Height
330

Left
180

Text


Top
180

Width
2655

Then create a command button and set the properties to:

Property
Value

Name
cmdPush

Caption
&Push

Default
True

Height
330

Left
180

Top
600

Width
1280

Then create a second command button and set the properties to:

Property
Value

Name
cmdPop

Caption
P&op

Height
330

Left
1560

Top
600

Width
1280

When you are done, your form should look like this:

The idea behind this user interface is that you type the plate type into the textbox, then press the Push button to add that plate to the stack. Pop removes the top plate on the stack and displays its contents in the textbox.

Select Format Lock Controls to keep the user interface from being able to be changed with the mouse from this point on.

Implementing Pushing a Plate on the Stack

You now need to implement the action of pushing a plate on the stack. You need something to represent a plate. To do this, you need to create a class module. In Visual Basic, perform these steps:

1. Use Project Add Class Module to create a new class module.

2. Select View Properties Window to display the properties window.

3. Set the Name Property of the class module to “pltPlate”

4. In the General Declarations section of the class module, add the following line of code:

Public strPlateType As String

A public variable in the General Declarations section of a class module defines a property of the class module. By declaring this variable, you defined the property strPlateType as a property of objects of type plate.

Now go back to the frmStack form and double-click on the Push command button. Make the event procedure look like this:

Private Sub cmdPush_Click()

    Dim pltNew As pltPlate

    Set pltNew = New pltPlate

    pltNew.strPlateType = txtPlateType.Text

End Sub

This code instantiates a new object from the pltPlate class and points the object variable pltNew at it. The object that pltNew refers to then has its strPlateType property set to the contents of the txtPlateType textbox.

The next step is to make the plate object remain in memory after the event procedure exits. As written, the code creates pltNew and sets the strPlateType property. However, because pltNew is a local variable with a procedure level lifetime (because it is declared with the Dim keyword), it goes out of scope when the End Sub is executed and is removed from memory. Each object maintains a count of how many items refer to the object, called its reference count. When the reference count goes to zero the object is automatically deleted from memory. Because pltNew is the only reference to the object, when it goes out of scope, the reference count on the plate goes to zero and the object is deleted. The way around this is to retain a reference to the plate object after the event procedure exits. To do this, add the following to the General Declarations section of the frmStack form:

Private pltTop As pltPlate

Then change the event procedure to look like this:

Private Sub cmdPush_Click()

    Dim pltNew As pltPlate

    Set pltNew = New pltPlate

    pltNew.strPlateType = txtPlateType.Text

    Set pltTop = pltNew

End Sub

Because pltTop is a private variable in the General Declarations section of the form, it lasts as long as the module for the form exists in memory. Before the procedure exits, pltTop is set to point to the same object that pltNew is pointing at, increasing the reference count to two. Now the reference count only falls to one and the object is maintained in memory when the event procedure exits and pltNew goes out of scope. A diagram of what is happening in memory will be provided a little later in the article.

The next step is to retain the plates previously pushed. As written, the code currently only maintains the most recent plate that was pushed. When pltTop is set to refer to the plate that was just created, it no longer points to the plate that it was referring to the last time that cmdPush was clicked. This causes the reference count on the previous plate to fall to zero, destroying it. You need to make each new plate maintain a reference to the plate that was previously the top plate on the stack to keep the reference count above zero and maintain the stack. To do this you need to add a property to the pltPlate class. Go to the pltPlate class and add a property to the General Declarations section, so that the entire contents of the pltPlate class looks like this:

Option Explicit

Public strPlateType As String

Public pltNext As pltPlate

The new property, pltNext, will be used to refer to the next object in the stack. If pltNext has the value Nothing (meaning that it does not point at anything), then this is the last plate in the stack. Now use that property in the cmdPush event procedure. Modify the code on the form so that the entire form module looks like this:

Option Explicit

Private pltTop As pltPlate

Private Sub cmdPush_Click()

    Dim pltNew As pltPlate

    Set pltNew = New pltPlate

    pltNew.strPlateType = txtPlateType.Text

    Set pltNew.pltNext = pltTop

    Set pltTop = pltNew

End Sub

Finally, after an item is pushed, empty the textbox and set the focus to it. Modify the code once more so that it looks like this:

Option Explicit

Private pltTop As pltPlate

Private Sub cmdPush_Click()

    Dim pltNew As pltPlate

    Set pltNew = New pltPlate

    pltNew.strPlateType = txtPlateType.Text

    Set pltNew.pltNext = pltTop

    Set pltTop = pltNew

    txtPlateType.Text = ""

    txtPlateType.SetFocus

End Sub

Save the project and name it stackui.vbp. Save the form and name it stack.frm. Save the class module and name it plate.cls. Run the code and test that you can type text into the textbox and press the Push command button without errors. Unless you use the debugger, you will not be able to determine if the code really works until you implement the Pop functionality.

Implementing Popping a Plate Off the Stack

The next step is to pop an item off the stack and display the plate type in the textbox. Double-click on the Pop command button and make the event procedure look like this:

Private Sub cmdPop_Click()

    txtPlateType.Text = pltTop.strPlateType

    Set pltTop = pltTop.pltNext

End Sub

The code, as written, works fine until you try to pop when there is nothing on the stack. When that happens, you underflow the stack. When Visual Basic tries to retrieve pltTop.strPlateType, pltTop is set to Nothing. Trying to retrieve a property of Nothing results in a run-time error 91 (Object variable or With block variable not set). To prevent that, change the code to look like this:

Private Sub cmdPop_Click()

    If Not (pltTop Is Nothing) Then

        txtPlateType.Text = pltTop.strPlateType

        Set pltTop = pltTop.pltNext

    End If

End Sub

Now try pushing and popping several plates until you are happy that the code works.

How the Pushing Code Works

To understand data structures, you should use pictures. Memory is organized like this when then program runs:

Figure 7: Organization of Memory when the Program Runs

Now type “Paper” into the textbox on the form and click the Push command button. Line by line, after the line:

Dim pltNew As pltPlate

This creates a new object variable, pltNew, but does not point it at anything. Memory looks like this, at this point:

Figure 8: Organization of Memory after the Dim of pltNew

After the line:

Set pltNew = New pltPlate

This creates a new object of type pltPlate with default values for its properties, then sets pltNew to point at the object. Memory looks like this, at this point:

Figure 8: Organization of Memory after Creating New Plate

After the line:

pltNew.strPlateType = txtPlateType.Text

Figure 9: Organization of Memory after filling pltNew.strPlateType

After the line:

Set pltNew.pltNext = pltTop

Notice that this does not do anything, because Nothing got copied over the value Nothing. Memory still looks like this at this point:

Figure 10: Organization of Memory After Setting pltNew.pltNext to the Value of pltTop

After the line:

Set pltTop = pltNew

This copies the pointer to the object from pltNew to pltTop and increases the reference count to two. Memory looks like this, at this point:

Figure 11: Organization of Memory After Setting pltTop to the Value of pltNew

After the line:

txtPlateType.Text = ""

This merely cleans out txtPlateType.Text. Memory looks like this:

Figure 12: Organization of Memory After Emptying Out txtPlateType.Text

After the line:

txtPlateType.SetFocus

This sets the focus to the textbox, which does not change memory at all.

Finally, after the line:

End Sub

The pltNew variable goes out of scope and is deleted. This causes the reference count to decrement. Memory then looks like this:

Figure 13: Organization of Memory After the End Sub

Now create your own diagrams for how memory looks when you type “Plastic” into the textbox and click the Push command button. After the End Sub line, memory should look like this:

Figure 14: Organization of Memory After Pushing Plastic

The Code for the Unencapsulated Stack

The code for the form frmStack should look like this:

Option Explicit

Private pltTop As pltPlate

Private Sub cmdPop_Click()

    If Not (pltTop Is Nothing) Then

        txtPlateType.Text = pltTop.strPlateType

        Set pltTop = pltTop.pltNext

    End If

End Sub

Private Sub cmdPush_Click()

    Dim pltNew As pltPlate

    Set pltNew = New pltPlate

    pltNew.strPlateType = txtPlateType.Text

    Set pltNew.pltNext = pltTop

    Set pltTop = pltNew

    txtPlateType.Text = ""

    txtPlateType.SetFocus

End Sub

The code for class module pltPlate should look like this:

Option Explicit

Public strPlateType As String

Public pltNext As pltPlate

Encapsulating the Stack

The problem with the implementation of the stack shown earlier is that the code for manipulating the stack mixes in with the code for the user interface. It would be better if you encapsulated the stack. Encapsulation is a term meaning that you place all of the code dealing with the implementation of the class inside the class definition.

You could encapsulate the stack in a standard module instead of a class module. However, further down the line, you will want to make the stack an ActiveX component. You should implement the interface with class modules since the interface of ActiveX components must be through class modules.

To encapsulate the stack:

1. Select Project Add Class Module to create a new class module.

2. Select View Properties Window to view the properties window.

3. Set the Name Property of the class module to “stkStack”

Now you need to add two methods to the stkStack class: Push and Pop. Methods are procedures that operate on the object. The way that you create a method is to create a public Sub or Function within the class module. In Visual Basic, follow these steps:

1. Select Tools Add Procedure to insert a new procedure into the class module.

2. Type the word Push into the Name textbox, leave the option buttons set to Sub and Public.

3. Press OK

4. Select Tools Add Procedure again from the menu

5. Type the word Pop into the Name textbox, but set the option button to “Function” this time.

6. Press OK.

The code should look like this:

Option Explicit

Public Sub Push()

End Sub

Public Function Pop()

End Function

Now modify the Push method so that it takes an argument and looks like this:

Public Sub Push(ByVal strPlateType As String)

End Sub

By default, the Pop method returns a value of the data type variant. Add a return type to change it to a string, like this:

Public Function Pop() As String

End Function

Then modify the entire module so that it looks like this (you can copy much of the code from frmStack):

Option Explicit

Private pltTop As pltPlate

Public Sub Push(ByVal strPlateType As String)

    Dim pltNew As pltPlate

    Set pltNew = New pltPlate

    pltNew.strPlateType = strPlateType

    Set pltNew.pltNext = pltTop

    Set pltTop = pltNew

End Sub

Public Function Pop() As String

    If Not (pltTop Is Nothing) Then

        Pop = pltTop.strPlateType

        Set pltTop = pltTop.pltNext

    End If

End Function

Finally, change the code in frmStack to look like this:

Option Explicit

Private stkPlates As New stkStack

Private Sub cmdPop_Click()

    txtPlateType.Text = stkPlates.Pop()

End Sub

Private Sub cmdPush_Click()

    stkPlates.Push txtPlateType.Text

    txtPlateType.Text = ""

    txtPlateType.SetFocus

End Sub

The New keyword in the line below causes a new object to be instantiated from the stkStack class the first time that stkPlates is referenced.

Private stkPlates As New stkStack

Notice that the new code in frmNew does not know anything about the implementation of the stack. It merely pushes things and pops things from the stack.

Handling Errors

The code, as currently written, does not do anything when you pop when there is nothing on the stack. What would you do if you wanted to do something different, such as putting up an error message? It is sometimes useful to generate errors within a class module and trap them in the calling module.

Rewrite the Pop procedure in the stkStack class module to look like this:

Public Function Pop() As String

    If pltTop Is Nothing Then

        Err.Raise Number:=vbObjectError + 5000, _

            Description:="Underflowed the stack"

    Else

        Pop = pltTop.strPlateType

        Set pltTop = pltTop.pltNext

    End If

End Function

The Err.Raise generates a runtime error with the given number and description. By adding the vbObjectError constant to a number that you invent for the error (in this case 5000), you guarantee the number does not conflict with any Visual Basic error.

Then change the code in cmdPop_Click in frmStack to look like this:

Private Sub cmdPop_Click()

    On Error GoTo ErrorHandler

    txtPlateType.Text = stkPlates.Pop()

ExitProcedure:

Exit Sub

ErrorHandler:

    Select Case Err.Number

    Case vbObjectError + 5000 'Underflowed the stack

        MsgBox Err.Description

    Case Else

        MsgBox "Unexpected runtime error" & vbCr _

            & "#" & Err.Number & " " & Err.Description

    End Select

    Resume ExitProcedure

End Sub

Investigating Properties of a Class Module

There are two ways to implement properties in a class module. You can create properties by creating public variables in the General Declarations section of the class module. There is another way to create properties: property procedures. There are at least three good reasons for creating property procedures:

· To create read-only or write-only properties.

· To create properties that restrict the value of the variables.

· To create properties that produce calculated values.

If you do not need any of this functionality, you should continue to use public variables in the General Declarations section of the class module.

Suppose that you want to produce a read-only Boolean property called boolEmpty that looks at the stack and returns true if the stack is empty. Open the stkStack class module and perform these steps:

1. Select Tools Add Procedure from the Menu

2. Type the boolEmpty into the Name textbox.

3. Select the Property option button.

4. Press OK.

Visual Basic inserts two procedures into your code: a Property Get and a Property Let. These look like this:

Public Property Get boolEmpty() As Variant

End Property

Public Property Let boolEmpty(ByVal vNewValue As Variant)

End Property

The Property Get is called any time the value of the property is retrieved. The Property Let is called any time the value of the property is set. By default, the data type of the property is a Variant. To change the data type to Boolean, you can change the code like this:

Public Property Get boolEmpty() As Boolean

End Property

Public Property Let boolEmpty(ByVal boolNewValue As Boolean)

End Property

Simply delete the Property Let procedure to make a property read-only. Delete the Property Get to make the property write-only. In this case, delete the Property Let to make boolEmpty read-only. Next, make boolEmpty show whether the stack is empty or not by changing the code to look like this:

Public Property Get boolEmpty() As Boolean

    boolEmpty = pltTop Is Nothing

End Property

Generally, in user interface design, it is better to keep the user from doing the wrong thing rather than telling the user “don’t do that” after the fact. You can use the boolEmpty property instead of the code you recently wrote to handle runtime errors to disable the Pop command button if the stack is empty.

Change the code in frmStack to look like this:

Option Explicit

Private stkPlates As New stkStack

Private Sub cmdPop_Click()

    txtPlateType.Text = stkPlates.Pop()

    cmdPop.Enabled = Not stkPlates.boolEmpty

End Sub

Private Sub cmdPush_Click()

    stkPlates.Push txtPlateType.Text

    txtPlateType.Text = ""

    txtPlateType.SetFocus

    cmdPop.Enabled = Not stkPlates.boolEmpty

End Sub

Private Sub Form_Load()

    cmdPop.Enabled = Not stkPlates.boolEmpty

End Sub

Code for the Encapsulated Stack

The code for frmStack should look like this:

Option Explicit

Private stkPlates As New stkStack

Private Sub cmdPop_Click()

    txtPlateType.Text = stkPlates.Pop()

    cmdPop.Enabled = Not stkPlates.boolEmpty

End Sub

Private Sub cmdPush_Click()

    stkPlates.Push txtPlateType.Text

    txtPlateType.Text = ""

    txtPlateType.SetFocus

    cmdPop.Enabled = Not stkPlates.boolEmpty

End Sub

Private Sub Form_Load()

    cmdPop.Enabled = Not stkPlates.boolEmpty

End Sub

The code for the stkStack class module should look like this:

Option Explicit

Private pltTop As pltPlate

Public Sub Push(ByVal strPlateType As String)

    Dim pltNew As pltPlate

    Set pltNew = New pltPlate

    pltNew.strPlateType = strPlateType

    Set pltNew.pltNext = pltTop

    Set pltTop = pltNew

End Sub

Public Function Pop() As String

    If pltTop Is Nothing Then

        Err.Raise Number:=vbObjectError + 5000, _

            Description:="Underflowed the stack"

    Else

        Pop = pltTop.strPlateType

        Set pltTop = pltTop.pltNext

    End If

End Function

Public Property Get boolEmpty() As Boolean

    boolEmpty = pltTop Is Nothing

End Property

The code for the pltPlate class module should look like this:

Option Explicit

Public strPlateType As String

Public pltNext As pltPlate

Making a Property Represent a Collection

Often it is useful to have a property that returns a collection of items stored by the object. This is especially true when you are creating complicated ActiveX component hierarchies, where a property returns a collection and each item in the collection represents another object that may also have properties that return collections. This section shows how to return a collection from an object. The collection reflects the entire contents of the stack in order.

A collection in VBA is a group of members that can be of any data type. While each member can be of a different data type, in practice each member is usually of the same data type. The code example will return a collection of strings.

To implement a read-only property that returns a collection of strings, follow these steps:

1. Open the stkStack class module.

2. Select Procedure from the Insert menu.

3. In the Name field of the dialog insert strsPlates.

4. Select the Property option button, and, of course, leave the procedure as Public.

5. Press OK.

6. Delete the Property Let procedure to make the property be read-only.

Now change the strsStack property procedure to look like this:

Public Property Get strsPlates() As Collection

    Dim strsTemp As New Collection

    Dim pltCur As pltPlate

    Set pltCur = pltTop

    Do Until pltCur Is Nothing

        strsTemp.Add pltCur.strPlateType

        Set pltCur = pltCur.pltNext

    Loop

    Set strsPlates = strsTemp

End Property

This code creates a collection, then walks the stack adding the plate type to the stack. Finally, it sets the return value of the property to refer to the collection.

Next, change the form and add a listbox. When you are done the properties for the form and the listbox should look like this:

Property
Value

Border Style
3-Fixed Dialog

Caption
Stack Example

Height
4020

Name
frmStack

Width
3120

Property
Value

Height
2205

Left
180

Name
lstStack

Top
1140

Width
2655

Then view the code and insert a procedure into the form called FillList. To do so, follow these steps:

1. Select Procedure from the Insert menu.

2. In the Name field type FillList.

3. In the Type frame select Sub.

4. In the Scope frame select Private.

Then change the GetStack sub to look like this:

Private Sub FillList()

    Dim strsPlates As Collection

    Dim istrsPlates As Long

    lstStack.Clear

    Set strsPlates = stkPlates.strsPlates

    For istrsPlates = 1 To strsPlates.Count

        lstStack.AddItem strsPlates.Item(istrsPlates)

    Next istrsPlates

End Sub

This code walks the collection and fills the listbox with the strings in the collection.

Finally add a line in the cmdPush_Click, cmdPop_Click, and Form_Load event handlers to Call FillList after any change to the stack has been made. This causes the listbox to always show the current contents of the stack.

The entire code for the form now looks like this:

Option Explicit

Private stkPlates As New stkStack

Private Sub cmdPop_Click()

    txtPlateType.Text = stkPlates.Pop()

    cmdPop.Enabled = Not stkPlates.boolEmpty

    Call FillList

End Sub

Private Sub cmdPush_Click()

    stkPlates.Push txtPlateType.Text

    txtPlateType.Text = ""

    txtPlateType.SetFocus

    cmdPop.Enabled = Not stkPlates.boolEmpty

    Call FillList

End Sub

Private Sub Form_Load()

    Call FillList

    cmdPop.Enabled = Not stkPlates.boolEmpty

End Sub

Private Sub FillList()

    Dim strsPlates As Collection

    Dim istrsPlates As Long

    lstStack.Clear

    Set strsPlates = stkPlates.strsPlates

    For istrsPlates = 1 To strsPlates.Count

        lstStack.AddItem strsPlates.Item(istrsPlates)

    Next istrsPlates

End Sub

Using the Initialize and Terminate Events

The Initialize event is the constructor for objects in VBA. It can be used to initialize the default values for objects. It can also be used for other initialization work, such as opening databases. The Terminate event is the destructor for object in VBA. Use the Terminate event to clean up things.

One drawback to using a linked list of objects to store the stack is that there is no persistence— VBA destroys the entire stack when the program is closed. If you want to make your stack persistent, you can write the information to a persistent data storage, such as a database, in the Terminate event. Then you can read the info in the Initialize event.

To implement persistence using a table in a database:

1. Create a database named stackui.mdb in the same directory as the stack project.

2. Create a table in the database named tblStack.

3. Open the table and add two fields, lngOrder of type long number, and strPlateType of type text.

4. Close the database.

Add the code below to stkStack class module.

Private Sub Class_Initialize()

    Dim db As Database

    Dim rst As Recordset

    Set db = DBEngine.Workspaces(0).OpenDatabase( _

        MakeNewFileName(App.Path, App.EXEName, "mdb"), _

        False, False)

    Set rst = db.OpenRecordset( _

        "select strPlateType from tblStack order by lngOrder desc;", _

        dbOpenDynaset)

    Do Until rst.EOF

        Push rst!strPlateType

        rst.MoveNext

    Loop

    rst.Close

    Set rst = Nothing

    db.Close

    Set db = Nothing

End Sub

Private Sub Class_Terminate()

    Dim db As Database

    Dim rst As Recordset

    Dim lngOrder As Long

    Set db = DBEngine.Workspaces(0).OpenDatabase( _

        MakeNewFileName(App.Path, App.EXEName, "mdb"), _

        False, False)

    db.Execute "delete * from tblStack;"

    Set rst = db.OpenRecordset("tblStack", dbOpenDynaset)

    lngOrder = 1

    Do Until boolEmpty

        rst.AddNew

        rst!lngOrder = lngOrder

        rst!strPlateType = Pop()

        rst.Update

        lngOrder = lngOrder + 1

    Loop

    rst.Close

    Set rst = Nothing

    db.Close

    Set db = Nothing

End Sub

Private Function MakeNewFileName(ByVal strPath As String, _

    ByVal strExeName As String, ByVal strExtension As String) As String

    Dim strFileName As String

    Dim istrFileName As Long

    If Right$(strPath, 1) = "\" Then

        strFileName = strPath & strExeName

    Else

        strFileName = strPath & "\" & strExeName

    End If

    istrFileName = Len(strFileName)

    Do Until Mid$(strFileName, istrFileName, 1) = "."

        istrFileName = istrFileName - 1

        If istrFileName = 0 Then

            strFileName = strFileName & "."

            istrFileName = Len(strFileName)

            Exit Do

        End If

    Loop

    MakeNewFileName = Left$(strFileName, istrFileName) & strExtension

End Function

Notice that because you encapsulated the stack, you do not have to change the user interface code at all.

Before the code will compile, you must select References from the Tools menu and ensure that Microsoft DAO Object library is checked.

Creating an ActiveX DLL

The next step is to take the stack and make it into a component that you can reuse. In Visual Basic 4.0, follow these procedures:

1. Create a new project.

2. Remove the default form by right clicking on Form1 in the Project window and selecting Remove File.

3. Add the stkStack and pltPlate classes by right clicking on the Project window and selecting Add File, then specifying the filename.

4. Open the stkStack class module and bring up the properties window.

5. Set the Instancing property to 2-Creatable Multiuse.

6. Set the Public property to True.

7. Select Module from the Insert menu.

8. Bring up the Properties window and name the module Globals.

9. Select Procedure from the Insert menu and create a new public sub named main.

10. Select Options from the Tools menu and move to the Project tab.

11. Ensure that the startup form says Sub Main.

12. Set the Project Name to StackX.

13. Set the Start Mode to OLE Server.

14. Set the Application Description to Stack ActiveX Component.

15. Press OK

16. Select References from the Tools menu.

17. Ensure that the Microsoft DAO Object Library is checked.

18. Select Make OLE DLL File from the File menu.

19. Name the file stackx.dll and press OK. This creates the DLL file on the disk.

20. Select Options from the Tools menu and move to the Project tab.

21. Click on the ... button next to the Compatible OLE Server field, select the stackx.dll file you just created, and press Open.

22. Press OK.

23. Save the project as stackx.vbp, and the module as globals.bas.

24. Rename the database to stackx.mdb using the Windows Explorer.

At this point, the stackx.dll ActiveX component is ready to use.

Changing the User Interface to Use StackX.DLL

Now change the user interface to use this component. Follow these steps:

1. Open stackui.vbp.

2. Right click on stack.cls in the Project window and select Remove File.

3. Right click on plate.cls in the Project window and select Remove File.

4. Select References from the Tools menu.

5. Uncheck Microsoft DAO Object Library.

6. Check Stack ActiveX Component.

7. Select OK.

You can now run the application and it should still work.

Summary

This text covered:

· A Word About Objects

· Terminology

· Advantages and Disadvantages of Object Oriented Programming

· When To Use Class Modules

· Data Structures

· Implementing a Stack Using Class Modules

· Creating the Interface

· Implementing Pushing a Plate on the Stack

· How the Pushing Code Works

· Implementing Popping a Plate Off the Stack

· The Code for the Unencapsulated Stack

· Encapsulating the Stack

· Handling Errors

· Investigating Properties of a Class Module

· Code for the Encapsulated Stack

· Using the Initialize and Terminate Events

· Creating an ActiveX Component (OLE DLL)

· Changing the User Interface to Use StackX.DLL

10

_914617901.vsd

_914634757.vsd

_914635683.vsd

_914636615.vsd

_914637104.vsd

_914637377.vsd

_914636350.vsd

_914634856.vsd

_914621493.vsd

_914634295.vsd

_914618604.vsd

_914616709.vsd

_914616896.vsd

_914616708.vsd

